- == [ЧАСТЬ 4] ==-

¦ Введение

Привет всем! Добро пожаловать в четвертую часть этого ряда тренеров!

Она немного запоздала, но я уверен, что Вы найдете, что ожидание стоило этого, потому что сегодня я собираюсь показывать Вам, как использовать очень мощный инструмент:

Виртуальные Экраны.

 =-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-= 

¦ Что такое - Виртуальный Экран и почему он нам необходим?

Позвольте нам предположить, что Вы генерируете многочисленные сложные экранные действия на лету (например, прокрутка экран, и тогда приходится менять все спрайты для каждого кадра игры, которую Вы пишете.)

Вы представляете себе, как ужасный выглядело бы, если пользователь мог бы фактически видеть стирание и изменение каждого спрайта для каждого кадра?

Вы могли бы представить нежелательное мерцание? Вы понимаете, что мог бы быть нежелательный эффект "удвоения спрайта" (где Вы видите две копии того же самого спрайта рядом друг с другом)?

В программе я включил часть, где я не использую виртуальные экраны, чтобы демонстрировать эти проблемы.

Виртуальные экраны - не единственный способ решить эти проблемы, но они - типично самые простые, для программирования.

Виртуальный экран - это секция памяти, которая является точным подобием VGA экрана. В ней вы делаете всю вашу работу, затем "зеркально отражаете" это на ваш истинный экран. В EGA 640x350x16 у вас автоматически есть виртуальная страница,

и можно получить до четырех в MCGA, используя часть видеопамяти, но для наших целей, мы используем основную память.

 =-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-= 

¦ Установка виртуального экрана

Как вы уже видели в первой части, экран MCGA объемом в 64000 байтов (320x200=64000).

Вы, возможно, также заметили, что в TP 6.0 вам недоступно больше памяти для нормальных переменных.

Например, высказывание

VAR Virtual : Array [1..64000] of byte;

будет "нет нет", поскольку у вас не будет места для других переменных.

Какое же решение? Я слышу многоголосый вопрошающий крик ваших умов. Ответ: указатели!

Указатели не используют основную 64КБ память, отведенную вам TP 6.0, а получают пространство где-нибудь еще в основной памяти 640КБ вашего компьютера.

Вот их установка:

Type Virtual = Array [1..64000] of byte;  { The size of our Virtual Screen }

     VirtPtr = ^Virtual;                  { Pointer to the virtual screen }

VAR Virscr : VirtPtr;                      { Our first Virtual screen }

    Vaddr  : word;                        { The segment of our virtual screen}

Если Вы поместите это в программу как есть, и попытаетесь использовать VirScr, ваша машина вероятно повиснет. Почему?

Потому что Вы должны получить память для ваших указателей прежде, чем Вы можете получить к ним доступ!

Это делается следующим образом:

Procedure SetUpVirtual;

BEGIN

  GetMem (VirScr,64000);

  vaddr := seg (virscr^);

END;

Когда Вы закончили вашу программу, стоит освободить память, захваченную виртуальным экраном, следующим образом:

Procedure ShutDown;

BEGIN

  FreeMem (VirScr,64000);

END;

Если Вы не сделаете это, ваши другие программы будут иметь меньшее количество памяти, чтобы использовать для себя.

 =-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-= 

¦ Рисование точки на вашем виртуальном экране

Это очень похоже на рисование точки на экране MCGA, как обсуждалось в части 1.

Вот наш оригинальный putpixel:

Procedure PutPixel (X,Y : Integer; Col : Byte);

BEGIN

  Mem [VGA:X+(Y*320)]:=col;

END;

Для нашего виртуального экрана, мы делаем следующее:

Procedure VirtPutPixel (X,Y : Integer; Col : Byte);

BEGIN

  Mem [Vaddr:X+(Y*320)]:=col;

END;

Кажется, весьма расточительным иметь две процедуры, делающие точно те же самые вещи, только на различных экранах, не так ли? Так почему бы не объединить подобно этому:

Procedure PutPixel (X,Y : Integer; Col : Byte; Where : Word);

BEGIN

  Mem [Where:X+(Y*320)]:=col;

END;

Чтобы использовать это, Вы будете писать кое-что вроде:

Putpixel (20,20,32, VGA);

PutPixel (30,30,64, Vaddr);

Эти две инструкции рисуют два пикселя ... один на экране VGA и один на виртуальном!

Не правда ли, это заставит вас прыгать от радости!? ;-) Вы заметите, что мы все еще не можем фактически ВИДЕТЬ виртуальный экран, так что вперед, к следующей части ...

 =-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-= 

¦ Как "перещелкнуть" ваш виртуальный экран на истинный экран

У вас фактически уже должны быть инструментальные средства, чтобы сделать это, непосредственно из информации полученной в предыдущих частях этого ряда тренеров.

Мы будем, конечно, использовать команду Move:

Move (Virscr^,mem [VGA:0],64000);

Просто, а? Вы может захотеть ждать verticle retrace (Часть 2) прежде, чем выполнить эту команду, поскольку это может сделать перещелкивание намного более гладким (и, увы, медленнее).

Обратите внимание, что большинство наших других процедур может быть изменено, чтобы поддержать виртуальный экран, типа Cls и т.д. (см. Часть 1 этого ряда), используя методы, описанные выше (я изменил процедуру CLS в программе, данной в конце этой Части.)

Мы - ASPHYXIA - использовали виртуальные экраны почти во всех наших демках.

Можете Вы вообразить, как ужасно SoftelDemo смотрел бы, если бы Вы были должны наблюдать изменяющийся перемещающийся фон, текст и vectorballs для КАЖДОГО кадра?

Нежелательное мерцание, эффекты удваивания и т.д. сделало бы это ужасным! Так что мы использовали виртуальный экран, и очень довольны результатом.

Обратите внимание, тем не менее, что, для того чтобы получить скорость необходимую нам, чтобы получить достаточно быструю демку, мы записали наши подпрограммы спрайтов, перещелкивания, палитры, и т.д. В АССЕМБЛЕРЕ. Команда перемещения очень быстра, но не так как АССЕМБЛЕР! ;-)

 =-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-=-= 

¦ Напоследок

Меня попросили сделать часть про прокрутку экрана, так что, вероятно, что я

сделать ее в течение следующей недели. Также, ASPHYXIA скоро выложит

маленькую демку с исходником на местных бордах. Она будет использовать подпрограммы, которые мы обсудили в этом ряде, и демонстрировать, как мощны они могут быть, если их использовать правильным способом.

Некоторые проекты для Вас:

1) Напишите зеркально отражающую инструкцию так, чтобы Вы могли писать:

flip (Vaddr, VGA); flip (VGA, Vaddr);

 (Это - то,как работает у ASPHYXIA)

2) Поместите большинство подпрограмм (putpixel, cls, pal и т.д.) в

модуль, так, чтобы не дублировать процедуры в каждой программе, которую вы пишете.

Увидимся на следующей неделе

                        - Denthor

